Documentation of a FORTRAN program, 'isocomp', for computing isostatic residual gravity

Robert C. Jachens

and

Carter W. Roberts

U.S. Geological Survey

Menlo Park, CA 94025

U.S. Geological Survey Open-File Report 81-574

Abstract

FORTRAN program 'isocomp' computes isostatic residual gravity from input data consisting of gravity station coordinates, complete Bouguer gravity and topography digitized on a geographic grid. It is based on the assumption of complete local isostatic compensation according to the Airy-Heiskanen system. The program requires numerical values for the assumed sea level crustal thickness, density of the topography, and density contrast across the Moho.

The program is designed for operation on the U.S. Geological Survey's Honeywell computer using the Multics operating system. As many as 2700 stations can be processed during a single run and the cpu time is approximately 3.2 seconds per station.

Preface

The program is designed for operation on the U.S. Geological Survey's Honeywell computer using the Multics operating system. The program has been extensively tested for stations located in North America where latitudes and longitudes are positive, so care should be exercised in the use of negative latitudes or longitudes. The maximum number of stations that can be input is 2700.

The program is written in Multics FORTRAN and is basically compatible with ANSI 66 FORTRAN. In transportation to another system, the calls to the Multics external subroutines io and close-file will need to be replaced by appropriate opening and closing file routines. In addition, most systems will require the program to be in upper-case.

The supplementary FORTRAN program, isoworld, (also documented in this report) for computing the combined isostatic and topographic corrections requires the Surface Gridding Library routines bgngrd, setgrd, clchgt, and endgrd for constructing a grid from the table and reading the interpolated values for the correction at each station. The Multics external subroutines asr, dsr, and close-file are used. The commands asr and dsr are used to locate the surface display routines on Multics.

Introduction

The interpretation of Bouguer gravity data in many areas is complicated by the presence of large, long-wavelength variations in the gravity field. On a regional scale, these long-wavelength variations tend to be strongly correlated with average elevation. Areas near sea level tend to have Bouguer gravity values near zero whereas areas characterized by high elevations, such as the Rocky Mountains and the Sierra Nevada Mountains, are associated with Bouguer gravity values that are strongly negative. In contrast, at sea where the earth's surface lies below sea level, Bouguer gravity tends to be strongly positive. Pronounced regional gradients in Bouguer gravity are associated with transition zones between areas of markedly different surface elevations, such as the transitions from ocean basins to continents or from the Great Plains to the Rocky Mountains.

It has long been recognized that this behavior of Bouguer gravity is a reflection of isostatic compensation within the earth (see Heiskanen and Vening Meinesz (1958) for a discussion of isostatic theories). In simplest terms, isostatic equilibrium means that below some relatively shallow depth within the earth, all stresses are hydrostatic regardless of whether they occur beneath mountains, lowlands, or ocean basins.

The gravity effects of isostatic compensation are so large in some areas that they tend to mask the gravity expressions of lateral density variations in the shallow crust, the targets of most gravity investigations. In the Klamath Mountains province of northern California and southern Oregon (Irwin, 1966) the gravity expression of numerous large thrust plates is superposed on a strong gravity gradient arising from the transition from oceanic to continental crust (Kim, 1974). In the Sierra Nevada Mountains of California, small gravity anomalies associated with individual granitic plutons are masked by the large gravity gradient arising from the crustal root beneath these mountains (Oliver and Robbins, in press).

In order to enhance the correlation between near-surface geology and gravity by separating the gravity effects of near-surface geologic structures from those of isostasy, a computer program (Appendix A) was written that computes the gravity effects of complete local isostatic compensation by the Airy-Heiskanen system (Heiskanen and Vening Meinesz, 1958, p. 135-137). The program requires topography digitized on a geographic grid.

Theory

The computer program is based on equations for the Airy-Heiskanen system given by Heiskanen and Vening Meinesz (1958, p. 135-137). Following their notation for a crust having a sea level thickness T, topography with an elevation above sea level h and a density ρ is underlain by a root of thickness t (fig. 1) satisfying the equation

$$t = (\rho / \Delta\rho)\, h \qquad (1)$$

where $\Delta \rho$ is the density contrast between the lower crust and upper mantle. Similarly, the thickness t' of the antiroot beneath the oceans is given by

$$t' = [(\rho - \rho_w)/\Delta\rho]\, h' \tag{2}$$

where ρ_w is the density of sea water and h' is the depth of the ocean (fig. 1). Therefore, the total crustal thickness, T_c, beneath moutains is

$$T_c = T + h + t \tag{3}$$

whereas total crustal thickness, T_o, beneath oceans is

$$T_o = T - h' - t' \tag{4}$$

The topography of the area for which the gravity effect of isostatic compensation is to be computed is divided into compartments and the average elevation in each compartment is estimated. The thickness of the root or antiroot beneath each compartment is determined from equation 1 or 2. The gravity effect at any observation point of the root or antiroot beneath each compartment is computed according to the following equations.

For areas above sea level

$$g_z = -\gamma \Delta\rho A \left\{ \left[r^2 + (T+e)^2\right]^{-\frac{1}{2}} - \left[r^2 + (T+e+t)^2\right]^{-\frac{1}{2}} \right\} \tag{5}$$

For areas below sea level

$$g_z' = \gamma \Delta\rho A \left\{ \left[r^2 + (T+e-t')^2\right]^{-\frac{1}{2}} - \left[r^2 + (T+e)^2\right]^{-\frac{1}{2}} \right\} \tag{6}$$

where

γ = the gravitational constant
A = area of the compartment
r = horizontal distance between the center of the compartment and the obervation point
e = elevation of the observation point

Equations 5 and 6 are essentially those of Heiskanen and Vening Meinesz (1958, p. 182, equations 6-33 and 6-34) modified slightly to include the elevation of the observation point. The quantities g_z and g_z' represent the gravity effect of a vertical mass-line located at the center of the root or antiroot compartment and having the total compensating mass of the compartment. Equations 5 and 6 do not take into account the curvature of the earth and, as a result, this program should be used only for compartments that are close enough to the observation point that the effects of curvature can be ignored. We generally use this approximation out to distances no greater than 166.7 km (Hayford Zone O). Published world-wide maps are available that give the combined effects of topography and compensation for all areas at distances greater than 166.7 km (Karki and others, 1961).

Test Cases

The program was tested against both an exact solution for a buried vertical cylinder and the published tables of Heiskanen (1938). The model parameters for the test against the exact solution were

$T = 25$ km
$\rho = 2.67$ g/cm^3
$\Delta\rho = 0.4$ g/cm^3
$h = 304.8$ m (uniform)
$e = 0$ km

In this test, the program yielded 18.40 mGal compared to an exact solution of 18.36 mGal. The computed area was 0.36% smaller than the area of the top surface of the cylinder.

A number of cases were computed and tested against the published tables and the results are shown in table 1. The common model parameters were

$\rho = 2.67$ g/cm^3
$\Delta\rho = 0.6$ g/cm^3
$h = 1800$ m (uniform)
$e = 0$ km
radius of computation = 58.8 km (through Hayford zone M)

Table 1

	isostatic effect (mGal)		area x 10^3 (km^2)	
T (km)	table	computed	exact	computed
20	-125.9	-124.94	10.86	10.69
30	-101.2	-100.14	10.86	10.69
40	- 81.2	- 80.26	10.86	10.69
60	- 53.5	- 52.64	10.86	10.69
*30	+ 71.6	+ 71.09	10.86	10.69

*antiroot computed from ocean depth of 1800 m

The slight discrepancies between the computed values and the test values probably result, in part, from slight differences in the computed and exact areas, the former in all test cases being slightly smaller than the latter. The computed area, in general, will be slightly different from the exact area because the outer boundary of the computation zone is not circular, but follows the boundaries of the 3x3 minute compartments included in the computation. All compartments for which the horizontal distance between the center of the compartment and the observation point is less than the specified radius of computations are included in the computation.

Another factor that contributes to the discrepancies between the computed and table values is that the table values were computed with the added constraint that the topographic and compensating masses were equal. This constraint results in slightly larger values for the table values (Heiskanen and Vening Meinesz, 1958, p. 136-137).

Program Usage

The program is designed for operation on the U.S. Geological Survey's Honeywell computer using the Multics operating system. It is interactive to the extent that during a remote terminal session, the program asks for all required input data either in the form of the data values or the names of files containing the data. For multi-station (>200) runs, the average cpu time per station is approximately 3.2 seconds (approximately $0.07 on Multics low priority queue). A maximum of 2700 stations can be processed during a single run.

Input

A sample terminal session is shown in Appendix C. The program asks for the names of two formatted input files. The first contains information about the observation points at which the gravity effect is to be calculated. One line (card image) is required for each observation and must include the following information (all numbers are right justified integers):

Column	data
1-8	station name
9-11	latitude (degrees, positive north)
12-15	latitude (minutes, to nearest hundredth)
16-19	longitude (degrees, positive west)
20-23	longitude (minutes, to nearest hundredth)
24-29	elevation (feet, to nearest tenth)
64-69	cBa1 (complete Bouguer anomaly at 2.67 g/cm^3, to nearest hundredth mGal)

The information in columns 30-63 is read in a34 format and carried through to the card image output files. These columns normally contain

Column	data
30-40	observed gravity (mGal, to nearest hundredth, with leading "9" deleted)
37-40	accuracy code
41-46	free-air anomaly

47-52	simple Bouguer anomaly (mGal, to nearest hundredth)
53-57	hand terrain correction (mGal, to nearest hundredth)
58-62	total terrain correction (mGal, to nearest hundredth)
63	zone through which hand terrain correction was carried

information in columns 70-80 is ignored

The format of these card images is identical to the card image output from Plouff's terrain correction program (Plouff, 1977).

The second file contain digitized elevation data. The data should be average elevations of 3x3 minute compartments arranged in 1x2 degree maps. The maps can be arranged in any order and must contain the following information:

<u>lead card</u>

columns	data
1-12	name of map
13-15	latitude of northwest corner of map (degrees, positive north)
16-18	latitude of northwest corner of map (minutes)
19-22	longitude of northwest corner of map (degrees, positive west)
23-25	longitude of northwest corner of map (minutes)
26	units of positive elevation (blank, zero or 1 = feet, 3 = meters. Add 4 to this number if compartments with negative elevations are filled with air rather than sea water)
27	units of negative elevations (blank, zero, or 1 = feet, 3 = meters, 6 = fathoms)

<u>80 cards</u> containing average elevations of 3x3 minute comparts (FORTRAN format 10F7.0).

The data are arranged in the standard form used for input to Plouff's terrain correction program. The first average elevation corresponds to the most northwestern 3x3 minute compartment of the map. Subsequent average elevations correspond to compartments progressing from west to east. Therefore, the first four "cards" contain average elevations of compartments in the most northern tier of the map. The next four "cards" contain average elevations of compartments in the next tier south and so on. Elevations can be in feet, meters, or fathoms, depending on the paramters contained on the lead "card" for each map. Examples of the two input files are given in Appendix C.

The program also asks for coordinate bounds (degrees) of the elevation data set required to perform the calculation, the radius to which the computed effect is desired (km), the sea level crustal thickness (km), density of the topography (g/cm^3), and the density contrast between the lower crust and upper mantle (g/cm^3). The coordinate bounds of the elevation data set are used simply to test whether entire maps should be ignored during computation. To process data for an entire 1° (n-s) x 2° (e-w) mid-latitude quadrangle to a distance of 166.7 km requires a coordinate range of 5° in latitude and 6° in longitude. For example, to process the Sacramento sheet (38-39°N, 120-122°W), the input would be 118 124 36 41 (fig. 2). All numerical data input from the terminal are read in variable format so that when more than one number is requested, they should be separated either by a blank or a comma.

Output

The program generates the following three output files: 1) a "print" file that gives information about the computations and isostatic corrections, 2) a "card image" output file that is identical to the input file with the addition of the isostatic residual gravity in columns 70-75 and the letters ISO in columns 76-78, and 3) a "save" file to permit data recovery following a system crash during program execution. Examples are given in Appendix C.

Supplemental program "isoworld"

For those who are working in the western United States (latitude 30-50°N, longitude 105-125°W) and have access to the Surface Gridding Library produced by Dynamic Graphics, Inc., an additional program, "isoworld" (Appendix B) is available that will take the "card" output file from "isocomp" and apply the combined isostatic and topographic corrections for all regions beyond 166.7 km. These corrections are obtained from Karki and others (1961). The output from this program is identical to the input with the following exceptions:

1) Columns 70-75, isostatic residual gravity including correction for regions beyond 166.7 km, and
2) a "W" in column 79 to indicate that "isoworld" has been used.

References

Heiskanen, W. A., 1938, New isostatic tables for the reduction of gravity values calculated on the basis of Airy's hypothesis: Publications Isostatic Institute of International Association of Geodesy, no. 2.

Heiskanen, W. A. and Vening Meinesz, F. A., 1958, The earth and its gravity field: McGraw-Hill, 470 p.

Irwin, W. P., 1966, Geology of the Klamath Mountains province, in Bailey, E. H., ed., Geology of northern California: California Division of Mines and Geology, Bulletin 190, p. 19-38.

Karki, P., Kivioja, L., and Heiskanen, W. A., 1961, Topographic-isostatic reduction maps for the world for the Hayford zones 18-1, Airy-Heiskanen system, T=30 km: Publications Isostatic Institute of International Association of Geodesy, no. 35.

Kim, C. K., 1974, A gravity investigation of the Weed sheet, northwestern California: Ph.D. thesis, University of Oregon, 145 p.

Oliver, H. W., and Robbins, S. L., 1981, Bouguer gravity map of California, Fresno sheet: California Division of Mines and Geology, (in press).

Plouff, D., 1977, Preliminary documentation for a FORTRAN program to compute gravity terrain correction based on topography digitized on a geographic grid: U.S. Geological Survey Open-File Report 77-535, 45 p.

APPENDIX A

LISTING OF PROGRAM "isocomp" AND SUBROUTINE "compcomp3"

****

```
c **** isocomp.fortran  updated 1/81 *****
c **** Isostatic corrections normally to radius of 166.7 km from each
c **** station calculated assuming complete local compensation by the
c **** Airy-Heiskanen system. Calculations performed using formulas
c **** of Heiskanen and Vening Meinesz (1958, "The earth and it's
c **** gravity field: McGraw-Hill Book Co) modified to incorporate
c **** station elevations. Crustal thickness calculated using
c **** topography averaged over 3x3 minute compartments.
c **** Input data is standard card output from Plouff terrain
c **** correction program. Terrain data is in standard 1x2 degree format
c **** used for terrain correction program.
c ****
c **** Note: This program  has only been tested for stations in
c **** North America where, following astronomical practice,all
c **** latitudes and longitudes are positive.

      external io(descriptors),close_file(descriptors)
       real mtr3,lat,lon,mtr
       character*50 aname,bname,cname,dname,ename
       character*12 mapname
       character*8 sta
      character*34 xx
       double precision gridlat,gcos,grlatstart
       dtr=1.7453293e-2
       mtr=2.90888e-4
       mtr3=8.72664e-4
      common sta(2700),selev(2700),g(2700),lat(2700),lon(2700),
     &elev(800),gsump(2700),asum(2700),sp(2700),sm(2700),ielev(2700),
     &xx(2700),lad(2700),lam(2700),lod(2700),lom(2700),ig(2700)

c ****** begin prompting section ******

       print 1
    1  format(" This program computes Airy-Heiskanen isostatic ",
     &"anomalies.")
       print 4
    4 format(1x,"Type the name of the gravity input file.")
       read 20, aname
       call io("attach","file12","vfile_",aname)
       print 5
    5 format(1x,"Type the name of the elevation digitization file.")
       read 20, bname
       call io("attach","file13","vfile_",bname)
       print 9
    9 format(1x,"Type in the minimum and maximum longitudes and the ",
     &"minimum and maximum"/1x,"latitudes of elevation block needed ",
     &"for computation.")
       read 20, lonmin,lonmax,latmin,latmax
```

```
      print 6
    6 format(" Type the name of the print output file")
      read 20,cname
      call io("attach","file14","vfile_",cname)
      print 7
    7 format(1x,"Type the name of the card image output file")
      read 20,dname
      call io("attach","file15","vfile_",dname)
      print 8
    8 format(1x,"Intermediate computational values will be saved in",
     &1x,"the event of a",/1x,"crash. Type the name of the file for ",
     &"this save.")
      read 20, ename
      call io("attach","file16","vfile_",ename)
      print 2
    2 format(" Type outer radius (km) of computation (usually 166.7)")
      read 20, outrad
   20 format(v)
      print 3
    3 format(1x,"Type the normal crustal thickness (km), density of ",
     &"the crust (g/cc)"/1x,"and the density contrast between crust",
     &1x,"and mantle (rho(m)-rho(c))")

c ************ end of prompting section ******

      areaexact=3.1415927*outrad**2
      orad=outrad*1.e5
      orad2=orad*orad
      read 20, tcrust,rho,deltarho
      alambda=rho/deltarho
      amu=(rho-1.027)/deltarho
      ac=6.673e-8*deltarho
      write(14,23)
   23 format(//"ISOSTATIC ANOMALY COMPUTATION",//"  Model Parameters",/)
      write(14,22) tcrust,rho,deltarho,outrad,areaexact
   22 format(1x,"CRUSTAL THICKNESS (km)",f10.2,/1x,"CRUSTAL DENSITY ",
     &"(g/cc)",f10.2,/1x,"DENSITY CONTRAST (g/cc)",f9.2,/1x,"OUTER ",
     &"RADIUS (km)",f15.2,/1x,"EXACT AREA (sq km)",f14.1)
      tcrust=tcrust*1.e5
c *****  read gravity data
c *********** begin reading station data ******
      do 110 i=1,2700
      read(12,30,end=98) sta(i),lad(i),lam(i),lod(i),lom(i),ielev(i),
     &xx(i),ig(i)
   30 format(a8,i3,3i4,i6,a34,i6)
      nnnn=i
      g(i)=ig(i)/100.
      lat(i)=lad(i)*dtr+lam(i)/100.*mtr
      lon(i)=lod(i)*dtr+lom(i)/100.*mtr
      clat=cos(lat(i))
      clat2=clat*clat
      sm(i)=(1.861656+clat2*(.000160*clat2-.019028))*1.e5
      sp(i)=(1.861656-.006343*clat2)*clat*1.e5
      tla=lat(i)*57.295780
```

```
      tlo=lon(i)*57.295780
          nsta=i
c ***** convert elev to cm since calculations done in cgs system *****
      selev(i)=ielev(i)*3.048
  110 continue
      write(14,24)nnnn
     print 24, nnnn
   24 format(/1x,"Computations will be made for",1x,i5,1x,"locations."/)
   98 continue
      do 361 i=1,nsta
      gsump(i)=0.
  361 asum(i)=0.
c *****  read elevation data by 2 degree sheets
      write(14,25)
   25 format(//"sequence of elevation data by quads",/)
  373 do 310 ll=1,200
     nnn=0
      read(13,15,end=1000) mapname,maplatd,maplatm,maplond,maplonm,
     &ifac1,ifac2
      grlatstart=maplatd*dtr+maplatm*mtr-4.36332e-4
      grlonstart=maplond*dtr+maplonm*mtr-4.36332e-4
   15 format(a12,2i3,i4,i3,2i1)
     if(maplatd.le.latmax.and.maplatd.gt.latmin.and.maplond.le.lonmax.
     &and.maplond.gt.lonmin) go to 374
      read(13,9783)  (elev(k),k=1,80)
 9783 format(f7.0)
      go to 373
  374 read(13,16,end=1000) (elev(k),k=1,800)
     write(14,33) mapname,maplatd,maplatm,maplond,maplonm,ifac1,ifac2
   33 format("elevations in map",1x,a12," latitude",2i4," longitude",
     &2i4," factor",1x,2i1)
     print 500, mapname
  500 format(1x,"Calculating effect of map",2x,a12)
   16 format(10f7.0)
c ***** Test for units of elevation data and convert to cgs .
     if(ifac1.eq.0.or.ifac1.eq.1.or.ifac1.eq.4.or.ifac1.eq.5)go to 320
     if(ifac1.eq.3.or.ifac1.eq.7)go to 321
     write(14,34)ifac1
     print 34,ifac1
   34 format(" Program does not recognize elevation factor ",i2,
     &". This map is ignored.")
     go to 310
  320 factp=30.48
     write(14,35)
   35 format(1x,"Elevations above sea level are in feet.")
     go to 325
  321 factp=100.0
     write(14,36)
   36 format(1x,"Elevations above sea level are in meters.")
  325 continue
     if(ifac2.eq.0.or.ifac2.eq.1)go to 322
     if(ifac2.eq.3)go to 323
     if(ifac2.eq.6)go to 324
     write(14,34)ifac2
```

```
      go to 310
  322 factm=30.48
      write(14,37)
   37 format(1x,"Elevations below sea level are in feet.")
      go to 326
  323 factm=100.0
      write(14,38)
   38 format(1x,"Elevations below sea level are in meters.")
      go to 326
  324 factm=182.88
      write(14,39)
   39 format(1x,"Elevations below sea level are in fathoms.")
  326 continue
      do 330 k=1,800
      if(elev(k).lt.0)go to 331
      elev(k)=elev(k)*factp
      go to 330
  331 elev(k)=elev(k)*factm
  330 continue
      kfac=1
      if(ifac1.eq.4.or.ifac1.eq.5.or.ifac1.eq.7)go to 345
      go to 346
  345 kfac=0
      write(14,347)
  347 format(1x,"Areas below sea level are filled with air.")
  346 continue
c ***** compute coordinates of grid center
       mcount=0
       do 360 mk=1,20
       gridlat=grlatstart-(mk-1)*mtr3
       gcos=dcos(gridlat)
       gcos2=gcos*gcos
       gm=(1.861656+gcos2*(.000160*gcos2-.019028))*1.e5
       gp=(1.861656-.006343*gcos2)*gcos*1.e5
       area=9*gm*gp
       do 360 nk=1,40
       mcount=mcount+1
       gridlon=grlonstart-(nk-1)*mtr3
c *** the next 3 lines account for negative elevations around Death
c *** Valley and other areas with air below sea level. Where this occurs
c *** the map lead card must have had 40 added to the index ie 73
c *** instead of 33 for elevations in meters.
       index=0
      if(elev(mcount).ge.0)go to 355
      index=1*kfac
   355 do 370 i=1,nsta
       d=lat(i)-gridlat
       dd=abs(d)*6400.
       if(dd.gt.outrad) go to 370
       arg2=gridlon-lon(i)
       x2=(0.5*(gp+sp(i)*(1+.25*d*d))*arg2/mtr)**2
       y2=(0.5*(gm+sm(i))*d/mtr)**2
       dist2=x2+y2
       if(dist2.gt.orad2) go to 370
```

```
c *****     compute gravity effect of grid element
      call compcomp3(alambda,amu,ac,dist2,area,tcrust,selev(i),
     &elev(mcount),index,f)
  390  gsump(i)=gsump(i)+f
       asum(i)=asum(i)+area
       grlat=gridlat*57.29578
       grlon=gridlon*57.29578
      if(elev(mcount).eq.0.)nnn=nnn+1
  370  continue
   360  continue
      if(nnn.ne.0)write(14,44)nnn
      if(nnn.ne.0)print 44, nnn
   44 format("***WARNING",i9,1x,"grid point/station combinations not ",
     &"calculated due to missing elevations. ***")
c ***** write save file *****
       rewind 16
       write(16,60) mapname
   60  format(1x,"partial sums through",2x,a12,"  quad")
       do 362 i=1,nsta
       gsp=gsump(i)*1000
       write(16,62) sta(i),gsp,asum(i)
   62 format(" sta  ",a8," gravity (equal pressure)",f8.2," area",e12.4)
  362  continue
  310  continue

 1000  write(14,26)
   26 format(//10x,"*********** SUMMARY ***********"/)
       tcrust=tcrust/100000
c *****compute total effect  and print out
      write(14,50)
   50 format(1x,"STATION",3x,"LATITUDE",2x,"LONGITUDE ELEVATION",2x,
     &"BOUGUER",4x,"ISOSTATIC",4x,"ISOSTATIC",5X,"AREA ERROR")
      write(14,51)
   51 format(2x,"NAME",5x,"DEG MIN   DEG  MIN",5x,"FEET",4x,"ANOMALY",
     &6x,"EFFECT",6,"ANOMALY",6x,"CALC-EXACT")
      write(14,52)
   52 format(43x,"MGALS",2x,"(EQUAL PRESSURE)",2x,"MGALS",6x,"SQ KM",
     &2x,"PERCENT"/)
       do 400 i=1,nsta
       gsumptot=0.
       areatot=0.
       gsumptot=gsumptot+gsump(i)*1000
  410  areatot=areatot+asum(i)
       gpcor=g(i)-gsumptot
      igpcor=gpcor*100
      flom=lom(i)/100.
      flam=lam(i)/100.
      felev=ielev(i)/10.
      aer=areatot*1.e-10-areaexact
      paer=aer*100/areaexact
c ****  write print output file ****
      write(14,53)sta(i),lad(i),flam,lod(i),flom,felev,g(i),gsumptot,
     &gpcor,aer,paer
   53 format(1x,a8,1x,i3,f6.2,1x,i4,f6.2,1x,f8.1,1x,f9.2,4x,f8.2,5x,
```

```
     &f8.2,3x,f8.1,f7.2)
c ***** write card format output file *****
      write(15,57)sta(i),lad(i),lam(i),lod(i),lom(i),ielev(i),xx(i),
     &ig(i),igpcor
   57 format(a8,1x,i2,i4,1x,i3,i4,i6,a34,2i6,"ISO")
  400  continue
       call close_file("-all")
       call io("detach","file12")
       call io("detach","file13")
       call io("detach","file14")
       call io("detach","file15")
       call io("detach","file16")
       stop
       end
c **** end of program isocomp ****


c *** subroutine compcomp3 for use with program isocomp ***
      subroutine compcomp3(alambda,amu,ac,dist,area,tcrust,sele,ele,
     &index,f)
       r=dist
       a=ac*area
       c=tcrust+sele
       if(index.eq.1) go to 5
       t=alambda*ele
       f=-a*(1./sqrt(r+c**2)-1./sqrt(r+(c+t)**2))
       go to 10
    5  t=-amu*ele
       f=a*(1./sqrt(r+(c-t)**2)-1./sqrt(r+c**2))
   10  continue
       return
       end
```

APPENDIX B
LISTING OF PROGRAM isoworld
****

```
c ***** isoworld.fortran
c     program isoworld modified 1/81 to take plouff format as input.
c      program to compute the isostatic correction beyond 166.7km
c      for all stations in the area 30 to 50 degrees north and 105 to
c      125 degrees west.
       dimension corr(2700),xdum(231),ydum(210),xxlon(2700),xxlat(2700)
       common sta(2700),com(2700),query(2700),xx(2700)
       dimension elev(21,21),ielev(2700),
     &ig(2700),lad(2700),lam(2700),lod(2700),lom(2700),iso(2700),
     &isow(2700)
       equivalence (xdum,elev(1,1)),(ydum,elev(1,12))
       character*8 sta
       character*34 xx
       character*50 aname,bname
       external close_file(descriptors),asr(descriptors),dsr(descriptors)
       data xdum/20.6,20.6,21.4,22.9,24.8,27.0,28.5,29.6,31,32,30.8,29.8,
     &28.9,26,23,20.8,18.2,16,14.9,13.2,12,
     &22.3,23,23.6,25,27,29.2,30.8,32.1,33.2,33.6,33,32,30.6,28.6,
     &25.2,22.6,20.3,18,16.6,14.8,13,
     &20,22.4,24.1,26,28,30.4,32,33.3,34.7,35.6,35,33.8,32.6,30.6,28,
     &25.1,22.2,19.6,17.5,16,14,
     &17.2,20.4,23.2,25,27,30,32.5,33.6,35,37.5,37,35.4,34.2,32.2,29.5,
     &27.4,23.7,21,18.6,17.2,15,
     &14.3,17.5,21,23,25,28.5,31.3,32.8,34.7,38,37.8,35.7,34.7,32.8,
     &30.3,28.2,24.7,22.2,19.6,17.6,15.8,
     &10.7,13.6,17.8,20.7,23,26,28.8,30.9,33,35.5,36,35.4,34.5,32.6,
     &30.8,28.4,25.2,23,20.6,18.7,16.8,
     &7.8,10.6,14,17.8,20.8,23,26,28.6,30.7,33,34.3,34.7,34.2,32.6,
     &31,29.5,25.8,24,21.8,20.2,18,4.6,7.2,10.4,14.7,17.7,20.3,23,26,
     &28.7,30.8,32.7,33.4,33.2,32.1,30.5,28.8,26.2,24.6,22.4,21,19.1,
     &1.4,4,7,10.4,15,18,21,23.7,26.6,29.2,30.8,31.6,31.5,30.6,28.5,
     &26.4,25,24,23,21.5,19.5,
     &-3,.6,4,7,11,15.2,19.3,22.1,24.5,27.9,29.4,30,29.9,27.6,26.2,
     &23.5,23,22.9,22.7,22,19.8,-8.4,-4,0,3.2,7,12,17.4,20.3,22.8,25.2,
     &28,28.3,27.5,25.9,24.6,21.9,21.5,21.6,22.2,22,19.8/
       data ydum/-12.2,-7.6,-4.2,-1,3,9,15,18.4,21,23,25,25.9,25.9,25,
     &24.1,21,20.4,20.3,20.5,21,19.8,
     &-17,-12.4,-7.5,-4.2,-0.5,5.8,12,16,19,21,22.6,23.9,24.1,23.8,
     &22,19.2,18,17.5,17.9,20.5,20.5,
     &-20.2,-16.8,-12.5,-7.7,-4,1,9.5,12.6,16.2,18.2,20.2,21.2,21.5,
     &21,19.3,17.3,16.1,15.7,15.9,18.4,20.7,
     &-24.5,-20.5,-16.6,-12.2,-8,-3,5,8.8,12,15.8,17.2,18.1,18.5,18.0,
     &16.8,15,13.7,13.2,13.8,15.2,19.4,
     &-27.2,-24.9,-21,-16.9,-13.2,-8.5,-.5,4.2,8,10.1,13,14.9,15.2,15.1,
     &14.1,12.3,11,10.6,10.9,12.8,16.6,
     &-29.6,-27.2,-25,-21.3,-17.2,-13.5,-9,-3,.3,4.5,7.2,9.5,12,12.3,
     &10.3,8.8,8,7.5,8.2,10.3,13.3,
     &-32,-30,-27.5,-25.2,-22,-17.8,-14.3,-10.8,-5.8,-3,-0.2,2.7,
     &4.8,6.7,6.2,5.1,4,4,5.2,8.5,10.8,
     &-33.5,-32.4,-30.4,-28.2,-26,-23,-19.5,-16.2,-13.8,-12,-7,-3.3,
```

```
     &-.3,.2,0,-.7,-1,0,3.3,5.3,8.6,
     &-34.2,-33.4,-32.2,-30.8,-29.2,-27,-24,-21,-18,-15,-11.3,-9,
     &-7.6,-5.2,-5.3,-5.7,-5.3,-3,0,4,6.3,
     &-34.6,-34.2,-33.6,-32.2,-31.6,-30.6,-29.2,-26,-23,-20,-16.6,
     &-14,-11.5,-10,-9.5,-8.8,-8,-6.8,-4.2,0,3.2/
      print 1
    1 format(1x,"Input file?")
      read(5,2)aname
    2 format(v)
      open(12,file=aname,form="formatted",mode="in")
      print 3
    3 format(1x,"Output file?")
      read (5,2)bname
      open(14,file=bname,form="formatted",mode="out")
      call asr(">iml>SDL","-after","working_dir")
      do 100 i=1,4000
      read(12,10,end=200)sta(i),lad(i),lam(i),lod(i),lom(i),ielev(i),
     &xx(i),ig(i),iso(i),com(i),query(i)
   10 format(a8,1x,i2,i4,1x,i3,i4,i6,a34,2i6,a3,1x,a1)
      xxlat(i)=lad(i)+(lam(i)/6000)-30.
      xxlon(i)=lod(i)+(lom(i)/6000)-105.
      ii=i
  100 continue
  200 continue
      call bgngrd
      call setgrd(21,0.,20.,21,0.,20.)
      call clchgt(xxlat,xxlon,ii,elev,21,21,corr)
      call endgrd
      do 400 k=1,ii
      isow(k)=iso(k)+(corr(k)*100)
      write(14,310)sta(k),lad(k),lam(k),lod(k),lom(k),ielev(k),xx(k),
     &ig(k),isow(k),com(k),query(k)
  310 format(a8,1x,i2,i4,1x,i3,i4,i6,a34,2i6,a3,"W",a1)
  400 continue
      call dsr(">iml>SDL")
      call close_file("-all")
      close(12)
      close(14)
      stop
      end
c *** end of program isoworld ***
```

APPENDIX C
INPUT/OUTPUT
****

```
*** SAMPLE ABSENTEE RUN OF PROGRAMS isocomp AND isoworld ***
    All responses of the user have been underlined.


 Absentee user CRoberts Calgrav logged in: 03/22/81  2016.4 pst Sun
 r 2016.5 $0.04 $0.04 1.207 1.207

isocomp
  This program computes Airy-Heiskanen isostatic anomalies.
  Type the name of the gravity input file.

test
  Type the name of the elevation digitization file.

california-dig
  Type in the minimum and maximum longitudes and the minimum and maximum
  latitudes of elevation block needed for computation.

120 126 36 41
  Type the name of the print output file

test-isoprint
  Type the name of the card image output file

test-out
  Intermediate computational values will be saved in the event of a
  crash. Type the name of the file for this save.

test-save
  Type outer radius (km) of computation (usually 166.7)

166.7
  Type the normal crustal thickness (km), density of the crust (g/cc)
  and the density contrast between crust and mantle (rho(m)-rho(c))

 25  2.67  0.40
  Calculating effect of map  REDING W1
  Calculating effect of map  REDDING
  Calculating effect of map  SUSANVILLE
  Calculating effect of map  UKIAH W1
  Calculating effect of map  UKIAH
  Calculating effect of map  CHICO
  Calculating effect of map  SNTA RO W1
  Calculating effect of map  SNTA ROSA
  Calculating effect of map  SACRAMENTO
  Calculating effect of map  SAN FRANCISC
  Calculating effect of map  SAN JOSE
  Calculating effect of map  OCEAN S CRUZ
```

```
 Calculating effect of map  SANTA CRUZ
 Calculating effect of map  SN FRNCSC W2

STOP
r 2016.9 $0.67 $0.70 21.006 22.214
```

<u>isoworld</u>

```
 Input file?
```

<u>test-out</u>

```
 Output file?
```

<u>test-iso</u>

```
-                  EXECUTION REPORT LOG
I            SURFACE GRIDDING LIBRARY (2.1C)
I            DYNAMIC GRAPHICS, INC.
I            DATE SUNDAY 22 MARCH 1981
I            TIME 8.16.56 P.M.
I            .
C    1 BGNGRD
C    2 SETGRD INMXCL=       21      XGDMIN= 0.00000     XGDMAX=  20.000
+             INMYRW=       21      YGDMIN= 0.00000     YGDMAX=  20.000
C    3 CLCHGT XPTARR= *ARRAY*       YPTARR= *ARRAY*     INMXYP=       3
+             ELVARR= *ARRAY*       IDMXCL=      21     IDMYRW=      21
+             HGTARR=*RESULT*
C    4 ENDGRD
S            .
S            SUMMARY OF EXECUTION
S            SGL TOTAL CALLS=         4
S            CALLS WITH ERRORS=       0
S            GRIDS CALCULATED=        0
S            TOTAL TIME IN SGL CALLS= 0.22000E-01 SECONDS.
S            (RESOLUTION OF TIMER IS SECS/1000.)
S            END OF SUMMARY

STOP
r 2016.9 $0.25 $0.95 1.002 23.216
```

<u>logout</u>

```
Absentee user CRoberts Calgrav logged out 03/22/81  2016.9 pst Sun
CPU usage 23 sec, memory usage 144.0 units
```

*** INPUT DATA SET TO PROGRAM isocomp ***

```
SFO C529 380000 1221919   1507997620F     -2646 -2697    0   48   -2650 -2983
SRO  701 385349 1234070  22208008218       2069  1312   24  411    1713 -1408
SRO W157 380248 1220009   5307998251      -2019 -2200   42   60   -2143 -1790
```

***** SAMPLE MAP FROM ELEVATION DIGITIZATION FILE *****

```
SAN FRANCISC 38      124     6
 -1980 -1950 -1910 -1900 -1830 -1670 -1400 -1130  -920  -670SANFRAN  1
  -330   -50   -50   -50   -60   -55   -54   -46   -36   -25SANFRAN  2
   -20   -22   -21    21   596   723   914   877   346   193SANFRAN  3
    78    -1     7   174   334   494   697   399   100    81SANFRAN  4
 -2000 -1970 -1920 -1900 -1850 -1750 -1560 -1300 -1170  -830SANFRAN  5
  -450   -80   -60   -60   -50   -52   -50   -48   -44   -37SANFRAN  6
   -34   -31   -27   -20    46   283   402  1341   989   150SANFRAN  7
    45     6    16    96   609   593   790   793   205   147SANFRAN  8
 -1980 -1930 -1880 -1830 -1800 -1770 -1700 -1450 -1330 -1000SANFRAN  9
  -650  -250   -70   -60   -50   -52   -49   -47   -46   -45SANFRAN 10
   -40   -35   -30   -25   -21   -11    -7   180   426   256SANFRAN 11
    84    28    -1     3   240  1016   791   590   386   371SANFRAN 12
 -1990 -1980 -1980 -1950 -1900 -1850 -1750 -1670 -1450 -1180SANFRAN 13
  -880  -550  -250   -80   -60   -44   -42   -45   -43   -41SANFRAN 14
   -39   -34   -30   -27   -23   -17   -14    -9    -3   112SANFRAN 15
    92    -6    -4     1    42   387   983   658  1122   798SANFRAN 16
 -2020 -2020 -1950 -1900 -1930 -1880 -1800 -1650 -1500 -1330SANFRAN 17
 -1100  -900  -700  -480  -230  -132   -50   -34   -35   -37SANFRAN 18
   -36   -31   -29   -28   -23   -19   -14    -7    -7     4SANFRAN 19
   211   145     1    -1     8    51   312   641   757   959SANFRAN 20
 -1960 -1950 -1910 -1880 -1840 -1800 -1770 -1730 -1640 -1380SANFRAN 21
 -1230 -1150  -950  -750  -600  -500  -200   -70   -40   -30SANFRAN 22
   -30   -30   -30   -30   -20   -19   -16   -12    -7     4SANFRAN 23
   256   261    28    -5    -1     1    28   280   477   731SANFRAN 24
 -1970 -1950 -1920 -1890 -1840 -1790 -1780 -1750 -1650 -1580SANFRAN 25
 -1500 -1350 -1180 -1000  -830  -700  -550  -250   -70   -30SANFRAN 26
   -30   -30   -30   -30   -20   -22   -18   -15   -13    -9SANFRAN 27
   286   361    26    -3    -2     1     3    34   189   714SANFRAN 28
 -1970 -1970 -1970 -1900 -1860 -1850 -1850 -1850 -1800 -1730SANFRAN 29
 -1550 -1300 -1200 -1170  -950  -750  -600  -500  -300   -80SANFRAN 30
   -60   -40   -40   -30   -30   -25   -22   -18   -15    -9SANFRAN 31
   408   226     2    -2    -3     1     1    10    31   465SANFRAN 32
 -1970 -1980 -2010 -2000 -1950 -1940 -1940 -1910 -1890 -1850SANFRAN 33
 -1670 -1350 -1150 -1030  -950  -850  -730  -600  -450  -200SANFRAN 34
   -70   -50   -50   -40   -30   -29   -26   -23   -17   114SANFRAN 35
   722   828   259    30    -1    -2     1     1    13    40SANFRAN 36
 -1970 -1950 -1930 -1900 -1800 -1750 -1700 -1650 -1600 -1550SANFRAN 37
 -1400 -1370 -1260 -1150 -1000  -830  -700  -550  -430  -250SANFRAN 38
  -100   -70   -50   -40   -40   -35   -32   -29   -25    28SANFRAN 39
   480   994   617   339    96     2     1     1    12    17SANFRAN 40
 -1970 -1950 -1930 -1900 -1800 -1750 -1700 -1650 -1600 -1400SANFRAN 41
 -1050 -1150 -1130  -980  -850  -720  -600  -520  -380  -280SANFRAN 42
  -150   -80   -60   -50   -40   -42   -37   -33   -29   -20SANFRAN 43
    -6   236   792   795   436    37    26     6     1     1SANFRAN 44
 -1980 -1970 -1900 -1850 -1800 -1750 -1700 -1650 -1600 -1400SANFRAN 45
 -1050 -1030 -1020  -950  -820  -600  -530  -450  -370  -320SANFRAN 46
  -250  -180  -100   -60   -50   -44   -41   -37   -32   -25SANFRAN 47
   -15   180   785  1659   861   283   156    38    12     5SANFRAN 48
 -1980 -1970 -1900 -1850 -1800 -1750 -1700 -1650 -1600 -1500SANFRAN 49
```

-1200	-650	-950	-1050	-850	-650	-550	-470	-450	-420SANFRAN	50
-350	-250	-100	-70	-50	-48	-45	-40	-35	-29SANFRAN	51
-17	27	488	1105	1498	818	579	272	133	92SANFRAN	52
-2050	-2000	-1880	-1850	-1830	-1800	-1700	-1650	-1600	-1500SANFRAN	53
-1200	-1000	-1250	-1120	-880	-700	-680	-600	-550	-430SANFRAN	54
-300	-220	-150	-80	-60	-55	-51	-47	-41	-34SANFRAN	55
-20	-7	366	558	794	1600	1836	1472	456	219SANFRAN	56
-2050	-2030	-2010	-1990	-1900	-1800	-1750	-1650	-1500	-1490SANFRAN	57
-1600	-1500	-1400	-1200	-1050	-830	-650	-500	-450	-400SANFRAN	58
-290	-220	-170	-100	-70	-59	-53	-50	-46	-40SANFRAN	59
-29	15	345	619	758	880	1550	1931	1097	430SANFRAN	60
-2080	-2060	-2050	-2000	-1900	-1850	-1800	-1750	-1700	-1600SANFRAN	61
-1650	-1600	-1550	-1350	-1100	-850	-700	-600	-450	-360SANFRAN	62
-290	-240	-180	-150	-100	-70	-57	-52	-49	-44SANFRAN	63
-32	16	250	703	1225	1606	1161	1786	2136	1494SANFRAN	64
-2080	-2060	-2050	-2010	-1930	-1870	-1800	-1750	-1700	-1650SANFRAN	65
-1780	-1720	-1500	-1250	-1080	-980	-880	-680	-450	-370SANFRAN	66
-340	-270	-230	-200	-160	-97	-62	-56	-53	-48SANFRAN	67
-39	-24	134	567	989	1300	1295	973	1420	1719SANFRAN	68
-2100	-2070	-2050	-2020	-1950	-1900	-1850	-1800	-1750	-1700SANFRAN	69
-1780	-1750	-1650	-1500	-1400	-1250	-1080	-830	-500	-450SANFRAN	70
-370	-320	-300	-260	-250	-193	-77	-60	-58	-54SANFRAN	71
-46	-35	-20	94	614	1336	1973	1064	897	1078SANFRAN	72
-2100	-2070	-2050	-2020	-2000	-1970	-1900	-1850	-1750	-1700SANFRAN	73
-1800	-1800	-1720	-1600	-1520	-1400	-1200	-1030	-850	-700SANFRAN	74
-530	-500	-400	-380	-400	-292	-148	-74	-62	-60SANFRAN	75
-56	-47	-36	-23	98	616	1391	1672	606	758SANFRAN	76
-2100	-2070	-2060	-2050	-2040	-2000	-1900	-1850	-1800	-1750SANFRAN	77
-1900	-1750	-1350	-1100	-1430	-1420	-1350	-1250	-1170	-1130SANFRAN	78
-1020	-900	-800	-670	-550	-349	-247	-161	-96	-68SANFRAN	79
-63	-72	-56	-41	-30	77	558	1027	758	531SANFRAN	80

*** PRINT OUTPUT FILE FROM PROGRAM isocomp ***

ISOSTATIC ANOMALY COMPUTATION

Model Parameters

CRUSTAL THICKNESS (km)	25.00
CRUSTAL DENSITY (g/cc)	2.67
DENSITY CONTRAST (g/cc)	0.40
OUTER RADIUS (km)	166.70
EXACT AREA (sq km)	87301.4

sequence of elevation data by quads

elevations in map REDING W1 latitude 41 0 longitude 126 0 factor 06
Elevations above sea level are in feet.
Elevations below sea level are in fathoms.
elevations in map REDDING latitude 41 0 longitude 124 0 factor 01

```
 Elevations above sea level are in feet.
 Elevations below sea level are in feet.
elevations in map SUSANVILLE   latitude  41    0 longitude 122    0 factor 01
 Elevations above sea level are in feet.
 Elevations below sea level are in feet.
elevations in map UKIAH W1     latitude  40    0 longitude 126    0 factor 06
 Elevations above sea level are in feet.
 Elevations below sea level are in fathoms.
elevations in map UKIAH        latitude  40    0 longitude 124    0 factor 01
 Elevations above sea level are in feet.
 Elevations below sea level are in feet.
elevations in map CHICO        latitude  40    0 longitude 122    0 factor 01
 Elevations above sea level are in feet.
 Elevations below sea level are in feet.
elevations in map SNTA RO W1   latitude  39    0 longitude 126    0 factor 06
 Elevations above sea level are in feet.
 Elevations below sea level are in fathoms.
elevations in map SNTA ROSA    latitude  39    0 longitude 124    0 factor 01
 Elevations above sea level are in feet.
 Elevations below sea level are in feet.
elevations in map SACRAMENTO   latitude  39    0 longitude 122    0 factor 01
 Elevations above sea level are in feet.
 Elevations below sea level are in feet.
elevations in map SAN FRANCISC latitude  38    0 longitude 124    0 factor 06
 Elevations above sea level are in feet.
 Elevations below sea level are in fathoms.
elevations in map SAN JOSE     latitude  38    0 longitude 122    0 factor 01
 Elevations above sea level are in feet.
 Elevations below sea level are in feet.
elevations in map OCEAN S CRUZ latitude  37    0 longitude 124    0 factor 06
 Elevations above sea level are in feet.
 Elevations below sea level are in fathoms.
elevations in map SANTA CRUZ   latitude  37    0 longitude 122    0 factor 06
 Elevations above sea level are in feet.
 Elevations below sea level are in fathoms.
elevations in map SN FRNCSC W2 latitude  38    0 longitude 126    0 factor 06
 Elevations above sea level are in feet.
 Elevations below sea level are in fathoms.
```

********** SUMMARY **********

STATION NAME	LATITUDE DEG MIN		LONGITUDE DEG MIN		ELEVATION FEET	BOUGUER ANOMALY MGALS	ISOSTATIC EFFECT (EQUAL PRESSURE)	ISOSTATIC ANOMALY MGALS	AREA ERROR CALC-EXACT SQ KM	PERCENT
SFO C529	38	0.00	122	19.19	15.0	-26.50	-5.06	-21.44	-60.8	-0.07
SRO 701	38	53.49	123	40.70	222.0	17.13	17.04	0.09	185.8	0.21
SRO W157	38	2.48	122	0.09	53.0	-21.43	-9.28	-12.15	207.3	0.24

*** CARD IMAGE OUTPUT FILE FROM isocomp - INPUT TO isoworld ***

```
SF0 C529 38    0 1221919   1507997620F     -2646 -2697    0   48   -2650 -2144ISO
SRO  701 385349 1234070  22208008218        2069  1312   24  411    1713     9ISO
SRO W157 38 248 122    9   5307998251      -2019 -2200   42   60   -2143 -1214ISO
```

*** SAVE FILE ***

```
partial sums through  SN FRNCSC W2  quad
sta  SF0 C529  gravity (equal pressure)   -5.06 area  0.8724E+15
sta  SRO  701  gravity (equal pressure)   17.04 area  0.8749E+15
sta  SRO W157  gravity (equal pressure)   -9.28 area  0.8751E+15
```

*** FINAL OUTPUT FILE FROM PROGRAM isoworld ***

```
SF0 C529 38    0 1221919   1507997620F     -2646 -2697    0   48   -2650 -2724ISOW
SRO  701 385349 1234070  22208008218        2069  1312   24  411    1713 -1371ISOW
SRO W157 38 248 122    9   5307998251      -2019 -2200   42   60   -2143 -1794ISOW
```



Figure 1. Isostasy--Airy-Heiskanen Model



Figure 2. Elevation data required for processing the Sacramento sheet.